Math 276
Calculus III

Dr. Constant J. Goutziers
Department of Mathematics, Computer Science and Statistics
goutzicj@oneonta.edu

Lesson 15

Maxima, Minima and Saddle Points

Initializations

```
> restart;
  with(plots):
  with(VectorCalculus):
  BasisFormat(false):
>
```

15.1 Maxima, Minima and Saddle Points

Examples

Example 15.1.1

Find the local maxima and minima of the function $f(x, y) = x\, y\, e^{-x^2 - y^2}$. Plot the surface showing all stationary points.

Solution
First we compute the stationary points.

```
> f:=(x, y)->x*y*exp(-x^2-y^2):
  f(x,y);
```

$$x\, y\, e^{-x^2 - y^2} \quad (2.1.1.1)$$

```
> for k to 2 do
  eq||k:=factor(D[k](f)(x,y)=0);
  od;
```

$$eq1 := -y\, e^{-x^2 - y^2} \left(-1 + 2\, x^2\right) = 0$$

$$eq2 := -x\, e^{-x^2 - y^2} \left(-1 + 2\, y^2\right) = 0 \quad (2.1.1.2)$$

```
> stps:=[solve({eq1, eq2}, {x, y})];
```

$$stps := \left[\{x = 0, y = 0\}, \left\{x = \mathrm{RootOf}\left(-1 + 2\, \_Z^2, label = \_L1\right), y = \mathrm{RootOf}\left(-1 + 2\, \_Z^2, label = \_L2\right)\right\}\right] \quad (2.1.1.3)$$

The **RootOf** is a placeholder for the solutions of the listed equation. To find those solutions we use the **allvalues** command.

```
> stps:=map(allvalues, stps);
```

(2.1.1.4)

$$stps := \left[\{x=0, y=0\}, \left\{y=\frac{1}{2}\sqrt{2}, x=\frac{1}{2}\sqrt{2}\right\}, \left\{y=\frac{1}{2}\sqrt{2}, x=-\frac{1}{2}\sqrt{2}\right\}, \left\{y=-\frac{1}{2}\sqrt{2}, x=\frac{1}{2}\sqrt{2}\right\}, \left\{y=-\frac{1}{2}\sqrt{2}, x=-\frac{1}{2}\sqrt{2}\right\} \right] \qquad \textbf{(2.1.1.4)}$$

We now use the second derivative test to determine which of these points correspond to a maximum, a minimum, or a saddle point.

```
> A:=factor(D[1$2](f)(x, y));
  B:=factor(D[2$2](f)(x, y));
  C:=factor(D[1,2](f)(x, y));
  discr:=factor(A*B-C^2);
```

$$A := 2\,x\,y\,e^{-x^2-y^2}\left(-3+2\,x^2\right)$$

$$B := 2\,x\,y\,e^{-x^2-y^2}\left(-3+2\,y^2\right)$$

$$C := e^{-x^2-y^2}\left(-1+2\,x^2\right)\left(-1+2\,y^2\right)$$

$$discr := -\left(e^{-x^2-y^2}\right)^2\left(-20\,x^2y^2+8\,x^2y^4+8\,x^4y^2+1-4\,y^2+4\,y^4-4\,x^2+4\,x^4\right) \qquad \textbf{(2.1.1.5)}$$

We print the stationary points information (SPI_k).

```
> for k to 5 do
  SPI[k]:=simplify(eval([discr, A, [x, y, f(x, y)]], stps
  [k]));
  od;
```

$$SPI_1 := [-1, 0, [0, 0, 0]]$$

$$SPI_2 := \left[4\,e^{-2}, -2\,e^{-1}, \left[\frac{1}{2}\sqrt{2}, \frac{1}{2}\sqrt{2}, \frac{1}{2}\,e^{-1}\right]\right]$$

$$SPI_3 := \left[4\,e^{-2}, 2\,e^{-1}, \left[-\frac{1}{2}\sqrt{2}, \frac{1}{2}\sqrt{2}, -\frac{1}{2}\,e^{-1}\right]\right]$$

$$SPI_4 := \left[4\,e^{-2}, 2\,e^{-1}, \left[\frac{1}{2}\sqrt{2}, -\frac{1}{2}\sqrt{2}, -\frac{1}{2}\,e^{-1}\right]\right]$$

$$SPI_5 := \left[4\,e^{-2}, -2\,e^{-1}, \left[-\frac{1}{2}\sqrt{2}, -\frac{1}{2}\sqrt{2}, \frac{1}{2}\,e^{-1}\right]\right] \qquad \textbf{(2.1.1.6)}$$

Point 1 is a saddle point, Point 2 and Point 5 are maxima, while Point 3 and Point 4 are minima.

Finally we plot the surface showing all stationary points.

```
> plot3d(f(x, y), x=-2..2, y=-2..2, style=patch, axes=
  boxed, labels=[x,y,z], orientation=[16, 66]);
```


z
0.12
0.02
-0.08
-0.18
y
-2
-1
0
1
2
x
-2
0
2

>